(* Asymptotic expected number of hairpins in ALL structures with theta=3 and stickiness p=1*)

(*We first compute the dominant singularity and asymptotic number of saturated structures, using Drmota−Lalley−Woods Theorem. Grammar with S, T where T means there is at least one base pair. *)

```
Clear["*"]
(* S arrow D or T; T arrow (E) or T* or (T) or S(D1) or S(T); D0 arrow * or D0 *;
D1 arrow *** or D1 *  *)
Clear[p, S, D0, D1, z, eqn0, eqn, F, z0, y0, dFdz0fz0S0, d2Fdy0fz0S0];

eqn0 = {S == z + z S + z^2 S + z^2 S^2}; (* grammar for all str, theta=1,p=1 *)
CellPrint["eqn0 is the usual (simple) grammar for all sec str, with theta=1,p=1"]
eqn1 = {S == D0 + T, T == z^2 D1 + z T + z^2 T + z^2 S T + z^2 S D1,
   D0 == z + z D0, D1 == z + z D1};
CellPrint[
 "eqn1 is variant of eqn,where D1 arrow * or D1 *, giving equivalent grammar to eqn0"]
Eliminate[eqn0, {D0, D1, T}]
Collect[%, S, Simplify]
Eliminate[eqn1, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["We derive the SAME equations
   from eqn0,eqn1 showing the underlying grammar is correct."]
p = 1;
eqn = {S == D0 + T, T == p z^2 D1 + z T + p z^2 T + p z^2 S T + p z^2 S D1,
   D0 == z + z D0, D1 == z^3 + z D1};
Eliminate[eqn, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["From previous expression, write S = ... to get F"]
F = (S^2 z^2 - z (-1 + z^2 + z^3)) / - (-1 + z + z^2 - z^3 - z^4);


NSolve[{F == S, D[F, S] == 1}, {z, S}]
z0 = 0.43691112721451186`;
y0 = 1.2887949921884878`;


dFdz0fz0S0 = D[F, z] /. {z → z0, S → y0}

d2Fdy0fz0S0 = D[F, {S, 2}] /. {z → z0, S → y0}
c = Sqrt[z0 dFdz0fz0S0 / (2 Pi d2Fdy0fz0S0)]
c * (1 / z0)^n n^(-3 / 2)
```

eqn0 is the usual (simple) grammar for all sec str, with theta=1,p=1

eqn1 is variant of eqn,where D1 arrow * or D1 *, giving equivalent grammar to eqn0

$$S^2 z^2 + S\left(-1 + z + z^2\right) == -z$$

$$S^2 z^2 + S\left(-1 + z + z^2\right) == -z$$

$S^2 z^2 + S(-1 + z + z^2) == -z$

$S^2 z^2 + S(-1 + z + z^2) == -z$

We derive the SAME equations from eqn0,eqn1 showing the underlying grammar is correct.

$S^2 z^2 + S(-1 + z + z^2 - z^3 - z^4) == z(-1 + z^2 + z^3)$

$S^2 z^2 + S(-1 + z + z^2 - z^3 - z^4) == z(-1 + z^2 + z^3)$

From previous expression, write S = ... to get F

```
{{z → -1.07833 - 1.4203 i, S → -1.33909 + 0.44663 i},
 {z → -1.07833 + 1.4203 i, S → -1.33909 - 0.44663 i},
 {z → 0.309017 + 0.951057 i, S → -1.30902 + 0.951057 i},
 {z → 0.309017 - 0.951057 i, S → -1.30902 - 0.951057 i}, {z → 0.436911, S → 1.28879},
 {z → -0.809017 + 0.587785 i, S → -0.190983 + 0.587785 i},
 {z → -0.809017 - 0.587785 i, S → -0.190983 - 0.587785 i}, {z → 0.71974, S → 0.389391}}
```

5.67452

0.775919

0.713121

$$\frac{0.713121 \times 2.28879^n}{n^{3/2}}$$

(*Now, we compute mean,variance using Drmota's Theorem*)

```
Clear["*"]
p = 1;
eqn = {S == D0 + T, T == p u z^2 D1 + z T + p z^2 T + p z^2 S T + p u z^2 S D1,
   D0 == z + z D0, D1 == z^3 + z D1};


CellPrint["Eliminate all variables except S,u,z"]
Eliminate[eqn, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["From previous expression, write S=... to obtain F"]
F = (S^2 (-1 + z) z^2 + z (-1 + z + z^2) - (u z^5 + S u z^5)) / - (1 - 2 z + 2 z^3);


f = (F /. S → s)
s - f
(* express over a common denominator*)
Together[s - f]
a = Numerator[%]
c = Denominator[Together[s - f]]
dfs = D[f, s]
1 - dfs
(* express over a common denominator*)
Together[1 - dfs]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
b = Numerator[%]
d = Denominator[Together[1 - dfs]]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
If[Denominator[Together[s - f]] == Denominator[Together[1 - dfs]],
```

```
 CellPrint["Denominator of (s-f) same as that of (1-dfs)"],
 CellPrint["Denominator of (s-f) different than than of (1-dfs)"]]
CellPrint["Now compute the resultant of numerators a,b to get relation between S,u,z"]

(* res =Resultant[s-f,1-dfs,s] *)


(*We compute resultant of numerators,
since we have Resultant[S-F,1-D[F,S],S]=0 which is not much use.*)
CellPrint["Express S-F and 1-D[F,S] as fractions over the same common denominator"]
CellPrint["Then compute the resultant of the numerators of these expressions"]
res = Resultant[a, b, s]
(* Replace z by z[u], a function of u *)


res /. z → z[u]
(* Now compute z'[u] *)
dres = D[%, u]
Simplify[Collect[dres, z'[u]]]
Solve[dres == 0, z'[u]]
dzu = Last[Last[Last[Solve[dres == 0, z'[u]]]]];
(*z[1] equals rho, the dominant singularity *)




rho = 0.43691112721451186`;
(* value of z0 in the first part of this file, the dominant singularity*)
dzuEvaluatedAt1 = (dzu /. u → 1) /. z[1] → rho
CellPrint[
 "According to Drmota's Theorem 1, the mean equals -z'[1]/z[1], computed next. "]

mu = ((-dzu / z[u] /. u → 1) /. z[1] → rho)

(*Variance computation *)
(* d2zu is z''[1] *)
d2zu = (((D[dzu, u] /. u → 1) /. z[1] → rho) /. z'[1] → dzuEvaluatedAt1)
CellPrint["Now compute variance, which by Drmota is -z''[1]/z[1] + mu^2 + mu"]
var = -d2zu / rho + mu * mu + mu
CellPrint["Now compute the standard deviation sigma"]
stdev = Sqrt[var]
```

Eliminate all variables except S,u,z

$$S - z - 2 S z + z^2 - S^2 z^2 + z^3 + 2 S z^3 + S^2 z^3 == (1 + S) u z^5$$

$$S^2 (-1 + z) z^2 + z \left(-1 + z + z^2\right) + S \left(1 - 2 z + 2 z^3\right) == u z^5 + S u z^5$$

From previous expression, write S=... to obtain F

$$\frac{s^2 (-1 + z) z^2 - u z^5 - s u z^5 + z \left(-1 + z + z^2\right)}{-1 + 2 z - 2 z^3}$$

$$s - \frac{s^2 (-1 + z) z^2 - u z^5 - s u z^5 + z \left(-1 + z + z^2\right)}{-1 + 2 z - 2 z^3}$$

$$\frac{s - z - 2 s z + z^2 - s^2 z^2 + z^3 + 2 s z^3 + s^2 z^3 - u z^5 - s u z^5}{1 - 2 z + 2 z^3}$$

$$s - z - 2 s z + z^2 - s^2 z^2 + z^3 + 2 s z^3 + s^2 z^3 - u z^5 - s u z^5$$

$$1-2z+2z^3$$

$$\frac{2s(-1+z)z^2-uz^5}{-1+2z-2z^3}$$

$$1-\frac{2s(-1+z)z^2-uz^5}{-1+2z-2z^3}$$

$$\frac{1-2z-2sz^2+2z^3+2sz^3-uz^5}{1-2z+2z^3}$$

$$1-2z-2sz^2+2z^3+2sz^3-uz^5$$

$$1-2z+2z^3$$

Denominator of (s-f) same as that of (1-dfs)

Now compute the resultant of numerators a,b to get relation between S,u,z

Express S-F and 1-D[F,S] as fractions over the same common denominator

Then compute the resultant of the numerators of these expressions

$$z^2-5z^3+8z^4-4z^5-2uz^7+6uz^8-8uz^9+4uz^{10}+u^2z^{12}-u^2z^{13}$$

$$z[u]^2-5z[u]^3+8z[u]^4-4z[u]^5-2uz[u]^7+6uz[u]^8-8uz[u]^9+4uz[u]^{10}+u^2z[u]^{12}-u^2z[u]^{13}$$

$$-2z[u]^7+6z[u]^8-8z[u]^9+4z[u]^{10}+2uz[u]^{12}-2uz[u]^{13}+2z[u]\text{ Derivative}[1][z][u]-15z[u]^2\text{ Derivative}[1][z][u]+32z[u]^3\text{ Derivative}[1][z][u]-20z[u]^4\text{ Derivative}[1][z][u]-14uz[u]^6\text{ Derivative}[1][z][u]+48uz[u]^7\text{ Derivative}[1][z][u]-72uz[u]^8\text{ Derivative}[1][z][u]+40uz[u]^9\text{ Derivative}[1][z][u]+12u^2z[u]^{11}\text{ Derivative}[1][z][u]-13u^2z[u]^{12}\text{ Derivative}[1][z][u]$$

$$-2z[u]^7+6z[u]^8-8z[u]^9+4z[u]^{10}+2uz[u]^{12}-2uz[u]^{13}+z[u]\left(2-15z[u]+32z[u]^2-20z[u]^3-14uz[u]^5+48uz[u]^6-72uz[u]^7+40uz[u]^8+12u^2z[u]^{10}-13u^2z[u]^{11}\right)\text{ Derivative}[1][z][u]$$

$$\left\{\left\{\text{Derivative}[1][z][u]\to-\left(2\left(z[u]^6-3z[u]^7+4z[u]^8-2z[u]^9-uz[u]^{11}+uz[u]^{12}\right)\right)\Big/\left(-2+15z[u]-32z[u]^2+20z[u]^3+14uz[u]^5-48uz[u]^6+72uz[u]^7-40uz[u]^8-12u^2z[u]^{10}+13u^2z[u]^{11}\right)\right\}\right\}$$

-0.0231775

According to Drmota's Theorem 1, the mean equals -z'[1]/z[1], computed next.

0.0530486

0.0172472

Now compute variance, which by Drmota is -z''[1]/z[1] + mu^2 + mu

0.0163873

Now compute the standard deviation sigma

0.128013